Mathematica 3

Lecture 14

For Today

- Interactivity
- Other types of boundary conditions
- Working with polar/spherical coordinates

Manipulate

The Manipulate command lets you interactively explore what happens when you vary parameters in real time:

```
In[1]:= Manipulate[Plot[Sin[f x], {x, −3, 3}, Filling → Axis], {f, 1, 5}]
```

Out[1]=




```
Manipulate[Plot[Sin[f * x + p], {x, -3, 3}, Filling → fill, PlotStyle → color],
 {f, 1, 5}, {p, 3, 9}, {fill, {Bottom, Top, Axis}}, {color, Red}]
```



Manipulate boundaries of PDE

```
Manipulate[
 sol = NDSolveValue[{-Laplacian[u[x, y], {x, y}] == 1, DirichletCondition[u[x, y] == 0., True]}, u,
   {x, y} ∈ RegionDifference[Rectangle[{0, 0}, {1, 1}], Rectangle[p1, p2]]];

 ContourPlot[sol[x, y], {x, y} ∈ sol["ElementMesh"], ColorFunction → "TemperatureMap"]

 , {{p1, {0.3, 0.2}}, Locator}, {{p2, {0.5, 0.4}}, Locator}]
```

$\{u, u_{min}, u_{max}\}$	manipulator (slider, animator, etc.)
$\{u, u_{min}, u_{max}, du\}$	discrete manipulator with step du
$\{u, \{x_{min}, y_{min}\}, \{x_{max}, y_{max}\}\}$	2D slider
{u, Locator}	a locator in a graphic
$\{u, \{u_1, u_2, \ldots\}\}$	setter bar for few elements; popup menu for more
$\{u, \{u_1 \to lbl_1, u_2 \to lbl_2, \ldots\}\}$	setter bar or popup menu with labels for elements
{u, {True, False}}	checkbox
{u, *color*}	color slider
{u}	blank input field
{u, FormObject[...]}	form with specified fields
{u, *func*}	create an arbitrary control from a function
$\{\{u, u_{init}\}, \ldots\}$	control with initial value u_{init}
$\{\{u, u_{init}, u_{lbl}\}, \ldots\}$	control with label u_{lbl}
{{u, ...}, ..., *opts*}	control with particular options
Control[...]	general control object
Delimiter	horizontal delimiter
string, view, cell expression, etc.	explicit text, view, cell, etc. annotations









Periodic Boundary Conditions

- Necessary for cyclic coordinates, e.g., ϕ in polar of spherical coordinates

- Useful for representing infinitely large systems

Specifying a Periodic Boundary

PeriodicBoundaryCondition[u[x_1, …], *pred*, *f*]
represents a periodic boundary condition $= u(x_{target}) = u(f(x_{target}))$ for all x_{target} on the boundary of the region given to NDSolve where *pred* is True.

PeriodicBoundaryCondition[a+b u[x_1, …], *pred*, *f*]
represents a generalized periodic boundary condition $a + b\,u(x_{target}) = u(f(x_{target}))$.

PeriodicBoundaryCondition is used together with differential equations to describe boundary conditions in functions such as NDSolve.

In NDSolve[*eqns*, {u_1, u_2, …}, {x_1, x_2, …}∈Ω], x_i are the independent variables, u_j are the dependent variables, and Ω is the region with boundary ∂Ω.



pred ~ defines part of the boundary which will be given from boundary conditions ("x_{target}")
f ~ the function which defines how u(x_{target}) relates to some other part of the boundary $u(x_{target}) = u(f(x_{target}))$

Cartesian Example (1D)



For $u(x)$ periodic in x:

pred → x==0 (sets x_{target} at left boundary)

f(x_) := x+L (defines value $u(x_{target}) = u(f(x_{target}))$

$$u(0) = u(L)$$

Alternately,

pred → x==L (sets x_{target} at right boundary)

f(x_) := x-L (defines value $u(x_{target}) = u(f(x_{target}))$

$$u(L) = u(0)$$

Free particle in 1-D with periodic boundary

```
L = 1;
f[x_] := x + L;
bcond = PeriodicBoundaryCondition[u[x], x == 0, f]
{vals, funcs} = NDEigensystem[{-Laplacian[u[x], {x}], bcond}, u[x], {x, 0, L}, 5]
Plot[funcs, {x, 0, L}]
```

```
PeriodicBoundaryCondition[u[x], x == 0, f]

{{0., 39.4789, 39.4789, 157.947, 157.947}, {InterpolatingFunction[ Domain: {{0., 1.}} Output: scalar ][x],
  InterpolatingFunction[ Domain: {{0., 1.}} Output: scalar ][x], InterpolatingFunction[ Domain: {{0., 1.}} Output: scalar ][x],
  InterpolatingFunction[ Domain: {{0., 1.}} Output: scalar ][x], InterpolatingFunction[
```



Kronig-Penny Model

PeriodicBoundaryCondition[a+b u[x_1, …], $pred$, f]
represents a generalized periodic boundary condition $a + b\,u(x_{target}) = u(f(x_{target}))$.

Bloch theorem:

$$T(\psi(x)) = e^{ika}\psi(x)$$

→ $\psi(x) = u(x)e^{ikx}$




```
b = .3; a = 1.0;
k = 0.0;
V2 = 1.0;
V[x_] := If[x > b, V2, 0];
{vals, funcs} =
  NDEigensystem[{-psi''[x] + V[x] * psi[x],
    PeriodicBoundaryCondition[psi[x] * Exp[-I k a], x == 0,
     TranslationTransform[{1}]]}, psi[x], {x, 0, a}, 5];
Plot[funcs[[2]], {x, 0, a}]
```



$k = \pi/2$

```
Plot[{Re[funcs[[2]]], Im[funcs[[2]]]}, {x, 0, a}]
```



$k = \pi/2$

```
Plot[Re[funcs[[2]] / Exp[I k x]], {x, 0, a}]
```



Formation of Energy Bands in solids

```
b = .3; a = 1.0;
k = Pi / 2;
V2 = 0.0;
V[x_] := If[x > b, V2, 0];
vals =
  Table[
   NDEigenvalues[{-psi''[x] + V[x] * psi[x],
     PeriodicBoundaryCondition[psi[x] * Exp[-I k a], x == 0, TranslationTransform[{1}]]},
    psi[x], {x, 0, 1}, 5], {k, -Pi / a, Pi / a, .05}];
ListPlot[Transpose[vals], Joined → True, PlotRange → {0, 100}]
```





2D Example

Alternately you could put,
`TranslationTransform[{4,0}]`

periodic in x, dirichlet in y

```
leqn = Laplacian[u[x, y], {x, y}] == -4 π ρ[x, y];
bcondx = PeriodicBoundaryCondition[u[x, y], x == -2, Function[r, r + {4, 0}]];
bcondy = DirichletCondition[u[x, y] == 0, (-2 < x < 2) && (y == -1 || y == 1)];
sol = NDSolveValue[{leqn, bcondx, bcondy}, u, {x, y} ∈ myregion];
ContourPlot[sol[x, y], {x, y} ∈ myregion, AspectRatio → Automatic]
```





x=0 x=L_x

periodicity in x:

$$pred \to x == 0$$
$$f(x\_) := x + L_x$$

Uniform charge density on disk

```
myregion = Rectangle[{-2, -1}, {2, 1}];
ρ[x_, y_] = If[(x^2 + y^2 < 1/4), 1, 0];
Show[RegionPlot[myregion, AspectRatio → Automatic], ContourPlot[ρ[x, y], {x, -2, 2}, {y, -1, 1}]]
```



I ran into trouble with periodic in x and y, simultaneously.

2D Example

Alternately you could put,
`TranslationTransform[{4,0}]`

periodic in x, dirichlet in y

```
leqn = Laplacian[u[x, y], {x, y}] == -4 π ρ[x, y];
bcondx = PeriodicBoundaryCondition[u[x, y], x == -2, Function[r, r + {4, 0}]];
bcondy = DirichletCondition[u[x, y] == 0, (-2 < x < 2) && (y == -1 || y == 1)];
sol = NDSolveValue[{leqn, bcondx, bcondy}, u, {x, y} ∈ myregion];
ContourPlot[sol[x, y], {x, y} ∈ myregion, AspectRatio → Automatic, PlotPoints → 100]
```





x=0 x=L_x

periodicity in x:

$$pred \to x == 0$$

$$f(x\_) := x + L_x$$

Uniform charge density on disk

```
myregion = Rectangle[{-2, -1}, {2, 1}];
ρ[x_, y_] = If[(x^2 + y^2 < 1/4), 1, 0];
Show[RegionPlot[myregion, AspectRatio → Automatic], ContourPlot[ρ[x, y], {x, -2, 2}, {y, -1, 1}]]
```



I ran into trouble with periodic in x and y, simultaneously.

Coordinate systems

Cartesian — Cartesian coordinate system

Cylindrical — cylindrical coordinate system

Spherical — spherical coordinate system

Paraboloidal ▪ ParabolicCylindrical ▪ EllipticCylindrical ▪ ProlateSpheroidal ▪ OblateSpheroidal

Bipolar — bipolar coordinate system

Bispherical — bispherical coordinate system

Toroidal — toroidal coordinate system

Conical ▪ ConfocalEllipsoidal ▪ ConfocalParaboloidal

```
Laplacian[f[r, θ, z], {r, θ, z}, "Cylindrical"] // Expand
```

$$f^{(0,0,2)}[r,\theta,z] + \frac{f^{(0,2,0)}[r,\theta,z]}{r^2} + \frac{f^{(1,0,0)}[r,\theta,z]}{r} + f^{(2,0,0)}[r,\theta,z]$$

```
Grad[f[r, θ, ϕ], {r, θ, ϕ}, "Spherical"]
```

$$\left\{f^{(1,0,0)}[r,\theta,\phi], \frac{f^{(0,1,0)}[r,\theta,\phi]}{r}, \frac{\mathrm{Csc}[\theta]\, f^{(0,0,1)}[r,\theta,\phi]}{r}\right\}$$

Solving a PDE in different coordinate systems

Cylindrical:

```
pvals = NDEigenvalues[
    {−Laplacian[u[r, ϕ], {r, ϕ}, "Polar"], DirichletCondition[u[r, ϕ] == 0, r == 1 && 0 < ϕ ≤ 2 π],
      PeriodicBoundaryCondition[u[r, ϕ], ϕ == 0, TranslationTransform[{0, 2 π}]]},
    u[r, ϕ], {r, 0, 1}, {ϕ, 0, 2 π}, 3]

{5.78319, 14.684, 14.684}
```

Cartesian:

```
avals = DEigenvalues[
    {−Laplacian[u[x, y], {x, y}], DirichletCondition[u[x, y] == 0, True]}, u[x, y], {x, y} ∈ Disk[], 3]
```

$\{\text{BesselJZero}[0, 1]^2, \text{BesselJZero}[1, 1]^2, \text{BesselJZero}[1, 1]^2\}$

```
pvals − avals
```

$\{6.9951 \times 10^{-6}, 0.00207181, 0.00207181\}$

changing coordinate systems

```
cSpherical[t_] := {1, t, 1 + 2 t + 3 t^2}
cCartesian[t_] = CoordinateTransform["Spherical" → "Cartesian", cSpherical[t]]
Show[Graphics3D[{Opacity[.25], Sphere[]}],
 ParametricPlot3D[cCartesian[t], {t, 0, 3}, ColorFunction → (Hue[.8 #4] &)]]
```

$\{\mathrm{Cos}[1 + 2t + 3t^2]\,\mathrm{Sin}[t],\ \mathrm{Sin}[t]\,\mathrm{Sin}[1 + 2t + 3t^2],\ \mathrm{Cos}[t]\}$



transforming points

```
CoordinateTransform["Cartesian" → "Spherical", {x, y, z}]
```

$\{\sqrt{x^2 + y^2 + z^2},\ \mathrm{ArcTan}[z, \sqrt{x^2 + y^2}],\ \mathrm{ArcTan}[x, y]\}$

```
Simplify[CoordinateTransform[{"Cartesian" → {"Toroidal", a}}, {x, y, z}]]
```

$\left\{\frac{1}{2}\,\mathrm{Log}\left[\frac{\left(a + \sqrt{x^2 + y^2}\right)^2 + z^2}{\left(a - \sqrt{x^2 + y^2}\right)^2 + z^2}\right],\ \mathrm{ArcTan}[-a^2 + x^2 + y^2 + z^2, 2 a z],\ \mathrm{ArcTan}[x, y]\right\}$

transforming fields

```
TransformedField["Cartesian" → "Spherical", {x, y, z},
  {x, y, z} → {r, θ, φ}] // Simplify
```

$\{r, 0, 0\}$

```
TransformedField[{"Cartesian" → {"Toroidal", 1}}, {x, y, z},
  {x, y, z} → {σ, τ, φ}] // Simplify
```

$\left\{\frac{\mathrm{Cos}[\tau]\,\mathrm{Sinh}[\sigma]}{\mathrm{Cos}[\tau] - \mathrm{Cosh}[\sigma]},\ \frac{\mathrm{Cosh}[\sigma]\,\mathrm{Sin}[\tau]}{\mathrm{Cos}[\tau] - \mathrm{Cosh}[\sigma]},\ 0\right\}$